
CHAPTER 8

Screen Processing 1:

Basic Features

Objective: To cover the requirements for displaying
information on a screen and accepting input from a keyboard.

INTRODUCTION

Up to this point, programs have defined data within an instruction operand
(immediate data) or initialized in the data area. The number of practical
applications for programs that process 0n1y defined data is few indeed.
Most programs require input data from a keyboard, disk, or modem, and
provide answers in a useful format on a screen, printer, or disk. Data for
screen and keyboard is in ASCII format.
The INT (Interrupt) instruction performs input and output. There are
various requirements for telling the system whether processing is to be
input or output, and on what device. This chapter covers basic requirements
for displaying information on a screen and for accepting input from a
keyboard.
You can perform all required screen/keyboard operations by the use of an
INT 1 OH instruction that transfers control directly to BIOS. However, to
facilitate some of the more complex operations, there is a higher level of
interrupt, INT 21H, that first transfers control to DOS. For example, input
from a keyboard may require a count of characters entered, a check against
a maximum number of characters, and a check for the return character. The
DOS INT 21H operation handles much of this additional processing and then
transfers automatically to BIOS.
The material in this chapter is suitable for both monochrome (black and
white, BW) and color video monitors. Chapters 9 and 10 cover more advanced
screen handling features and the use of color.

THE INTERRUPT INSTRUCTION: INT

The INT instruction interrupts processing, transfers to DOS or to BIOS for
specified action, and returns to the program to continue processing Most
often, an interrupt is to perform an input or output operation All
interrupts require a trail for exiting from your program and for returning
For this purpose, INT performs the following:

� Decrements the stack pointer by 2 and pushes the flags register onto the
stack.

� Clears the TF and IF flags.
� Decrements the stack pointer by 2 and pushes the CS register onto the

stack.



� Decrements the stack pointer by 2 and pushes the instruction pointer onto
the stack.

� Causes the required operation to be performed.
� Pops the registers off the stack and returns to the instruction following

the INT.

This process is entirely automatic, and your only concern is to define a
stack segment large enough for the necessary pushing and popping.

The two types of interrupts that this chapter covers are BIOS operation INT
10H for screen and keyboard processing and DOS operation INT 21 H for
displaying output and accepting input. For the latter, operations use what
are known as function calls to request an action Depending on requirements,
examples use both INT 10H and INT 21H

SETTING THE CURSOR

The screen is a grid of addressable locations at any one of which the
cursor can be set. A typical video monitor, for example, has 25 rows
(numbered 0 to 24) and 80 columns (numbered 0 to 79). Some examples of
cursor locations follow:

Decimal Format Hex Format

Location Row Column Row Column
Upper left corner 00 00 00 00
Upper right corner 00 79 00 4F
Center of screen 12 39/40 0C 27/28
Lower left corner 24 00 18 00
Lower right corner 24 79 18 4F

Features of the INT 1 OH instruction include setting the cursor at any
location and clearing the screen. The following sets the cursor to row 05
and column 12:

MOV AH,02 ;Request set cursor
MOV BH,00 ;Screen #O
MOV DH,05 ;Row O5
MOV DL,12 ;Column 12
INT 10H ; Interrupt -- exit to BIOS

The value 02 in the AH notifies INT 10H to set the cursor. Row and column
must be in the DX register, and screen (or "page") number, normally 0, is
in the BH. The content of the other registers is not important. To set row
and column, you could also use one MOV instruction with an immediate hex
value as

MOV DX,050CH ;Row 5, column 12

CLEARING THE SCREEN

Prompts and commands stay on the screen until overwritten or scrolled off.
When your program starts executing, you may want the screen cleared. You
can dear beginning at any location and ending at any higher numbered
location. Insert the starting row/column in the DX, the value 07 in the BH,



and 0600H in the AX. The following clears the entire screen:
MOV AX,0600H ;AH 06 (scroll), AL 00 (full screen)
MOV BH,07 ;Normal attribute (black & white)
MOV CX,0000 ;Upper left row/col
MOV DX,184FH ;Lower right row/col
INT 10H ;Interrupt -- exit to BIOS

The value 06 in the AH notifies INT 10H to scroll. This operation scrolls a
full screen to blank; the next chapter describes scrolling in more detail.
If you mistakenly set the lower right location higher than hex 184F, the
operation wraps around the screen and clears some locations twice. Although
the action causes no harm on mono screens, it may cause a serious error on
some color monitors.

SCREEN AND KEYBOARD OPERATIONS: ORIGINAL DOS

A program often has to display messages indicating completion or errors
detected and prompts to a user requesting data or action to take. We will
first examine the methods for original DOS versions, and later examine the
extended method introduced by DOS 2.0. The original DOS operations work
under all versions, although the DOS manual recommends use of the extended
operations for new development. The display operation for original DOS is a
little more involved, but keyboard input is easier to use because of its
built-in checks.

DISPLAYING ON THE SCREEN: ORIGINAL DOS

Displaying under original DOS requires defining a prompt message in the
data area, setting the AH register to 09 (a DOS function call), and issuing
a DOS INT 21H instruction. The operation recognizes the end of a message by
a dollar sign ($) delimiter, as shown next:

NAMPRMP DB 'Customer name? ', '$'
MOV AH,09 ;Request display
LEA DX,NAMPRMP ;Load address of prompt
INT 21H ;DOS interrupt

You can code a dollar sign delimiter immediately following the prompt as
just shown, inside the prompt as 'Customer name? $', or on the next line as
DB '$'. The effect, however, is that you can't use this operation to
display a $ character on the screen. Also, if you omit the dollar sign at
the end of the display string, the operation displays characters from
memory until it finds one - if any.
The LEA instruction loads the address of NAMPRMP in the DX to enable DOS to
locate the information that is to display. For the actual memory address,
LEA loads the offset address of NAMPRMP, and DOS uses the address in the DS
register plus the DX (DS:DX).

PROGRAM: DISPLAYING THE ASCIE CHARACTER SET

Most of the 256 ASCII characters are represented by a symbol that can
display on a video screen. Hex 00 and FF have no symbol and display as
blank, although the true ASCII blank character is hex 20.
The COM program in Figure 8-1 displays the entire range of ASCII-
characters. The program calls three procedures: B10CLR, Cl0SET, and
Dl0DISP. B10CLR clears the screen and Cl0SET initializes the cursor to



00,00.  Dl0DISP displays the contents of CTR, which is initialized to hex
00 and is successively incremented for each display until reaching hex FF.

page 60,132
TITLE ALLASC (COM)  Display ASCII characters 00-FF
CODESG SEGMENT PARA 'Code'

ASSUME CS:CODESG,DS:CODESG,SS:CODESG,ES:NOTHING
ORG 100H

BEGIN: JMP SHORT MAIN
CTR DB 00,’$’
; Main procedure:
MAIN PROC NEAR

CALL B10CLR ; Clear screen
CALL Cl0SET ; Set cursor
CALL D10DISP ; Display chars
RET

MAIN ENDP
; clear screen:

B10CLR  PROC
MOV AX,0600H
MOV BH,07
MOV CX,0000   ;upper left location
MOV DX,184FH   ;Lower right location
INT 10H
RET

B10CLR  ENDP
; Set cursor to 00,00:

Cl0SET  PROC
MOV AH,02
MOV BH,00
MOV DX, 0000
INT 10H
RET

Cl0SET  ENDP
; Display ASCII characters:

D10DISP PROC
MOV CX,256 ;Initialize for 256 iterations
LEA DX,CTR ;Initialize address of count

D20:
            MOV AH,09 ;Display ASCII char
            INT 21H
            INC CTR ; Increment count
            LOOP D20 ; Decrement CX, loop if nonzero
            RET ; Terminate
D10DISP ENDP

CODESG  ENDS
END BEGIN

Figure 8-1 Displaying the ASCII Character Set.

A problem is that the dollar symbol does not display and the characters
between hex 08 and hex OD, which are special "forms control characters" for
backspacing and so forth, cause the cursor to move Suggestion: Reproduce



the program as it stands, assemble it, link it and convert it to a COM
file. To run, enter its name, such as B:ASCII.COM
The first displayed line begins with a blank character (hex 00) two "happy
faces" (hex 0l and 02), and then a heart, a diamond, and a club (hex 03,
04, and 05). Hex 07 causes the speaker to sound. Hex 06 would have been a
spade but the control characters hex 08 through OD erased it. In fact, hex
0D caused a "carriage return" to the start of the next line. The musical
note is hex 0E. The characters above hex 7F include graphics symbols.
You can change the program to bypass the control characters. The following
instructions bypass all characters between hex 08 and OD; you may want to
experiment with bypassing only, say, hex 08 (backspace) and 0D (carriage
return).

CMP CTR,0BH ;Lower than OB?
JB D30 ;Yes -- accept

CMP  CTR,0DH ;Lower/equal OD?
JBE  D40 ;Yes -- bypass

D30:
MOV AH,40H ;Display < 08 and > OD
INT 21H

D40:
INC CTR

ACCEPTING INPUT FROM THE KEYBOARD: ORIGINAL DOS

The procedure to accept data from a keyboard is similar to that for
displaying output. For input using original DOS, the input area requires a
parameter list containing specified fields that the INT operation is to
process. First, the interrupt needs to know the maximum length of the input
reply. The purpose is to warn users who key in a reply that is too long;
the operation sounds the speaker and does not accept additional characters.
Second, the operation returns the length in bytes of the actual reply into
the parameter list.
The following defines a parameter list for an input area. LABEL is a
pseudo-op with the type attribute of BYTE. The first byte is your limit for
the maximum length of input. Since this is a one-byte field, the maximum is
hex FF, or 255. The second byte is for DOS to store the actual number of
characters entered. The third byte begins a field that is to contain the
typed characters.

MAMEPAR LABEL BYTE ; Start of parameter list:
MAXLEN DB 20 ; Max length of input
ACTLEN DB ? ; Actual length
NAMEFLD DB 20 DUP('  ') ;Chars entered from keyboard

In the parameter list, since the LABEL pseudo-op takes no space, NAMEPAR
and MAXLEN refer to the same memory location. MASM can also use the STRUC
pseudo-op to define a parameter list as a structure. However, since
references to names defined within a structure require special addressing,
our discussion delays this topic until Chapter 24, "Assembler Pseudo-op
Reference."
To request input, use DOS function call 10 (hex 0AH) in the AH, load the
address of the parameter list (NAMEPAR in the example) into the DX, and
issue INT 21H as follows:

MOV  AH,0AH ;Request input function
LEA  DX,NAMEPAR ;Load address of param list
INT  21H ;DOS interrupt



The INT operation waits for a user to enter characters and checks that the
number entered does not exceed the maximum in the parameter list (20 in the
example). The user presses the return key (hex OD) to signal the end of an
entry. This character also enters the input field (NAMEFLD in the
examples). If you key in a name such as BROWN (Return), the parameter list
appears as follows:

decimal:   |20| 5| B| R| O| W| N| #|  |  |  |  |...
hex:       |14|05|42|52|4F|57|4E|0D|20|20|20|20|...

The operation delivers the length of the input name, 05, into the second
byte of the parameter list, named ACTLEN in the example. The return
character is at NAMEFLD+5. The # symbol here is to indicate this character
because hex OD has no printable symbol. Since the maximum length of 20
includes the hex OD, the actual name may be only 19 characters long.

PROGRAM: ACCEPTING AND DISPLAYING NAMES

The EXE program in Figure 8-2 requests that a user enters a name, then
displays the name on the center of the screen and sounds the speaker. The
program continues accepting and displaying names until the user presses
Return as a reply to a prompt. Consider a user entering the name
TED SMITH:

TITLE   CTRNAME (EXE) Accept names & center on screen

STACKSG SEGMENT PARA STACK ‘Stack’
DW 32 DUP(?)

STACKSG ENDS

DATASG  SEGMENT PARA  Data
NAMEPAR LABEL BYTE ;Name parameter list:
MAXNLEN DB 20 ;max. length of name
NAMELEN DB ? ;no. chars entered
NAMEFLD DB 20 DUP('  '),'$' ;name, & delimiter

;for displaying
PROMPT  DB 'Name? ', '$'
DATASG  ENDS

CODESG  SEGMENT PARA  'Code'
BEGIN   PROC FAR

ASSUME  CS:CODESG,DS:DATASG,SS:STACKSG,ES:DATASG
PUSH DS
SUB AX,AX
PUSH AX
MOV AX,DATASG
MOV DS,AX
MOV ES,AX
CALL  Q10CLR ;Clear screen

A20LOOP:
MOV DX,0000 ;Set cursor to 00,00
CALL Q20CURS
CALL  B10PRMP ;Display prompt
CALL D10INPT ; provide for input of name
CALL Q10CLR ; clear screen
CMP NAMELEN,00 ;Name entered?
JE A30 ;    no --  exit



CALL El0CODE ; Set bell & '$'
CALL Fl0CENT ; Center & display name
JMP A20LOOP

A30:
RET ; Return to DOS

BEGIN   ENDP
; Display prompt:

B10PRMP PROC NEAR
MOV AH,09 ;Request display
LEA DX,PROMPT
INT 21H
RET

B1OPRMP ENDP
; Accept input of name:
D10INPT PROC NEAR

MOV AH,0AH ;Request input
LEA DX,NAMEPAR
INT 21H
RET

D10INPT ENDP

; Set bell  & '$' delimiter:
El0CODE PROC NEAR

MOV BH,00 ;Replace return char <OD)
MOV BL,NAMELEN ;with bell  (07)
MOV NAMEFLD[BX],07
MOV NAMEFLD[BX+1],'$' ;Set display delimiter
RET

E10CCDE ENDP
; Center & display name:
Fl0CENT PROC NEAR

MOV DL,NAMELEN ;Locate center column:
SHR DL,1   ;divide length by 2,
NEG DL  ;reverse sign,
ADD DL,40   ;add 40
MOV DH,12 ;Center row
CALL Q20CURS ; Set cursor
MOV AH,09
LEA DX,NAMEFLD ;Display name
INT 21H
RET

Fl0CENT ENDP
; clear screen:
Ql0CLR PROC NEAR

MOV AX,0600H ;Request scroll screen
MOV BH,30 ;Color (07 for BW)
MOV CX,0000 ;From 00,00
MOV DX,1B4FH ;To 24,79
INT 10H ; Call BIOS
RET

Q10CLR ENDP
; Set cursor row/col:
Q20CURS PROC NEAR ; DX set on entry

MOV AH,02 ;Request set cursor
MOV BH,00 ;Page #0
INT 10H ; Call BIOS
RET

Q2OCURS ENDP



CODESG  ENDS
END BEGIN

Figure 8-2 Accepting and Displaying Names

1. Divide the length 09 by 2 = 4, and

2. Subtract this value from 40 = 36

In E10CENT, the SHR instruction shifts the length 09 one bit to the right,
effectively dividing it by 2. Bits 00001001 become 00000100. The NEG
instruction reverses the sign, changing +4 to -4. ADD adds the value 40,
giving the starting position for the column, 36, in the DL register. With
the cursor set at row 12, column 36, the name appears on the screen as
follows:

Row 12: TED SMITH
              |  |
Column: 36  40

Note the instruction in El0CODE that inserts the bell (07) character in the
input area immediately following the name:

MOV  MAMEFLD[BX],07

The preceding instruction set the BX with the length. The effect of this
MOV is to combine the length in the BX with the address of NAMEFLD and to
move the 07 to this address. Thus, for a length of 05, the instruction
inserts 07 at NAMEFLD+05 (replacing the return character) following the
name. The last instruction in El0CODE inserts a delimiter following the 07
so that when Fl0CENT displays the name, the speaker also sounds.

Entering Only the Return Character

If you key in a name that exceeds the maximum in a parameter list, the
speaker sounds and the operation accepts only the return character. But if
you key in only Return, the operation accepts it and inserts a length of
zero in the parameter list, as follows:

Parameter list (hex): |14|00|0D|...

To signify end of input data on a prompt for name, a user can simply press
Return. If the length is zero, the program determines that input is ended.

Clearing the Return Character

You can use an input value for various purposes, such as printing on
reports, storing in a table, or writing on disk. For these purposes, you
may have to replace the return character (hex 0D) somewhere in NAMEFLD with
a blank (hex 20). The field containing the actual length NAMELEN, provides
its relative position. If NAMELEN contains 05 then this position is
NAMEFLD+5. You can move this length into the BX register for indexing the
address of NAMEFLD as follows:



MOV  BH,00 ;Set BX
MOV  BL,NAMELEN ;  to 00 05
MOV  NAMEFLD[BX],20H  ;Clear ret char to blank

The third MOV instruction moves a blank (hex 20) to the address specified
in the first operand: the address of NAMEFLD plus the contents of BX, in
effect, NAMEFLD+5.

Clearing the Input Area

Entered characters replace the previous contents in an input area, and
remain there until other characters replace them. Assume the following
successive input:

Input NAMEPAR (hex)
1. BROWN |14|05|42|52|4F|57|4E|0D|20|20|20|...|20|
2. HAMILTON |14|08|4B|41|4D|49|4C|54|4F|4E|0D|...|20|
3, ADAMS |14|05|41|44|41|4D|53|0D|45|5A|0D|...|20|

The name HAMILTON replaces the shorter name BROWN. But because the name
ADAMS is shorter than HAMILTON, it replaces HAMIL and the return character
replaces the T. The remaining letters, ON, still follow ADAMS. A useful
practice is to clear NAMEFLD prior to prompting for a name as follows:

MOV CX,20 ;Initialize for 20 loops
MOV SI,0000 ;Start position for name

B30:
MOV NAMEFLD[SI],20H ;One blank to name
INC SI ;Increment for next char
LOOP B30 ; 20 times

Instead of the SI register, you could use DI or BX. A more efficient method
that moves a word of two blanks requires only ten loops. However, because
NAMEFLD is defined as DB (byte), you would have to override its length with
a WORD and PTR (pointer) operand as the following indicates:

MOV CX,l0 ; Initialize for 10 loops
LEA SI,NAMEFLD ; Initialize start of name

B30:
MOV WORD PTR[SI],2020H ; Two blanks to name
INC SI ; Increment 2 positions
INC SI ; in name
LOOP B30 ; Loop 10 times

Interpret the MOV at B30 as: Move a blank word to where the address in the
SI register points. This example uses LEA to initialize and uses a slightly
different method for the MOV at B30 because you cannot code an instruction
such as

MOV WORD PTR[NAMEFLD],2020H ;Invalid

Clearing the input area solves the problem of short names being followed by
previous data. A more efficient practice is to clear only positions to the
right of an entered name.

SCREEN AND KEYBOARD OPERATIONS: EXTENDED DOS

We'll now examine the extended method introduced by DOS 2.0, which is more
in the UNIX style. If you use a version of DOS prior to 2.0, you won't be
able to execute the programs in this section. The extended method involves



a file handle that you set in the BX register when requesting an
I/O-operation. The following standard file handles are always available:

0 Input, normally keyboard (CON)
1 Output, normally display (CON)
2 Error output, display (CON)
3 Auxiliary device (AUX)
4 Printer (LPT1 or PRN)

The DOS interrupt is INT 21H, and the required function call is requested
in the AH register: hex 3F for input and hex 40 for output. Set the CX with
the number of bytes to read or display, and load the DX with the address of
the input or output area.
A successful operation clears the carry flag and inserts in the AX the
number of characters actually entered or displayed. An unsuccessful
operation sets the CF flag and inserts an error code (6 in this case) in
the AX. Since the AX could contain either a length or an error code, the
only way to determine an error condition is to test the CF flag, although
keyboard and display errors would presumably be rare. You also use file
handles in a similar way to process disk files, where error conditions are
more common.
You can use these function calls to redirect input and output to other
devices, although we won't concern ourselves with this feature here.

DISPLAYING ON THE SCREEN: EXTENDED DOS

The following instructions illustrate the use of the extended DOS function
required to request display:

DISAREA DB 20 DUP('  ') ;Display area
...
MOV AH,40H ;Request display
MOV BX,01 ;Fi1e handle for output
MOV CX,20 ;Maximum 20 characters
LEA DX,DISAREA ;Display area
INT 21H ; Call DOS

LEA loads the address of DISAREA in the DX to enable DOS to locate the
information that is to display. A successful operation clears the carry
flag (which you may test) and sets the AX with the number of characters
displayed. An unsuccessful operation could occur because of an invalid
handle. The operation sets the CF flag and inserts an error code (6 in this
case) in the AX. Since the AX could contain either a length or an error
code, the only way to determine an error condition is to test the CF flag.

Exercise: Displaying on the Screen

Let's use DEBUG to examine the internal effects of an interrupt. Load
DEBUG, and when its prompt appears, type A 100 in order to begin entering
assembler instructions (but not the numbers) at location 100. Remember that
DEBUG assumes that all numbers entered are hexadecimal.

100 MOV AH,40
102 MOV BX,01
105 MOV CX, xx   (Insert length of your name)
108 MOV DX,10E
l0B INT 21
l0D RET
10E DB 'Your name'



The program sets the AH to request display and sets the hex value 10F in
the DX - the location of the DB containing your name, at the end of the
program.
When you have keyed in the instructions, press Return again. Try the U
command (U 100,10D) to unassemble the program and R and then repeat T
commands to trace execution. When DEBUG executes INT 21H, it traces through
BIOS, so on reaching 10B, use the GO command (G 10D) to execute directly
through to the next instruction. Your name should display on the screen.
Use the Q command to return to DOS.

ACCEPTING INPUT FROM THE KEYBOARD: EXTENDED DOS

The following illustrates use of the extended DOS function to request
keyboard input:

INAREA DB 20 DUP(' ') ;Input area
…
MOV AH,3FH ;Request input
MOV BX,00 ;File handle for keyboard
MOV CX,20 ;Maximum 20 characters
LEA DX,INAREA ;Input area
INT 21H ; Call DOS

LEA loads the offset address of INAREA in the DX. The INT operation waits
for the user to enter characters, but does not check if the number of
characters exceeds the maximum in the CX register (20 in the example).
Pressing the return key (hex OD) signals the end of an entry. For example,
entering the characters "PC Users Group" causes the following in INAREA:

PC Users Group, hex 0D, hex 0A

The entered characters are immediately followed by a carriage return (hex
0D), which you entered, and a line feed (hex 0A), which you did not enter.
Because of this feature, the maximum number and the length of the input
area should provide for an additional two characters. If you enter fewer
characters than the maximum, the locations in memory following the entered
characters still contain the previous contents.
A successful operation clears the CF flag (which you may test) and sets the
AX with the number of characters delivered. In the preceding example, this
length is 14, plus 2 for the return and line feed, or 16. Accordingly, a
program can determine the actual number of characters entered. Although
this feature is trivial for YES and NO type of replies, it is useful for
replies with variable length such as names.
An unsuccessful operation could occur because of an invalid handle. The
operation sets the CF flag and inserts an error code (6 in this case) in
the AX. Since the AX could contain either a length or an error code, the
only way to determine an error condition is to test the CF flag.
If you key in a name that exceeds the maximum in the CX register, the
operation accepts all characters. Consider a situation in which the CX
contains 08 and a user enters the characters "PC Exchange." The operation
sets the first eight characters in the input area as "PC Excha" with no
return and line feed following, and sets the AX with a length of 08. The
next INT operation does not accept a name directly from the keyboard,
because it still has the rest of the previous string in its buffer. It
delivers "nge" followed by the return and line feed to the input area and
sets the AX to 05. Both operations are "normal" and the CF flag is cleared.

First INT: PC Excha AX = 08



Second INT: nge, 0D, 0A AX = 05

A program can tell that a user has keyed in a valid number of characters if
(a) the number returned in the AX is less than the number in the CX or (b)
the number returned is equal and the last two characters in the input area
are 0D and 0A.
The built-in checks of original DOS function call OAH for keyboard input
offer a far more powerful operation and, at least at the time of this
writing, that operation is the preferred choice.

Exercise: Entering Data

Here's an exercise in which you can view the effect of entering data while
in DEBUG. The program allows you to enter up to 12 characters, including
return and line feed. Load DEBUG, and when the prompt appears, type A 100
to begin entering assembler instructions at location 100. Remember that
DEBUG assumes that numbers entered are hexadecimal.

100 MOV AH,3F
102 MOV BX,00
105 MOV CX,0C
108 MOV DX,10F
10B INT 21
10D JMP 100
10F DB  ‘ ‘

The program sets the AH and BX to request keyboard input, inserts the
maximum length in the CX, and sets hex 10F in the DX - the location of the
DB at the end of the program. The entered characters appear beginning at
location hex 10F.

When you key in the instructions, press Return again. Try the U command
(U 100,108) to unassemble the program. Use R then repeated T commands to
trace execution of the four MOV instructions. When at location 10B, use G
10D to execute through the interrupt (don 't follow it through BIOS). DEBUG
stops to let you enter characters, and is followed by Return. Check the
contents of the AX register, the carry flag, and use D 10F to display the
entered characters in memory. You can continue looping indefinitely. Enter
Q to terminate any time.

USE OF CARRIAGE RETURN, LINE FEED, AND TAB FOR DISPLAY

One way to make displays more efficient is to use the carriage return, line
feed, and tab characters. You can code them in ASCII or hex, as:

ASCII Hex
CR 13 0DH
LF 10 0AH
TAB 09 09H

Use these characters where you display or accept input for advancing the
cursor automatically to the start of the next row, either for original or
extended DOS. Example:

MESSAGE DB 09,’PC Users Group Annual Report‘ , 13, 10
MOV AH,40H ;Request display
MOV BX,01 ;Handle



MOV CX,31 ;Length
LEA DX,MESSAG1 ;Message
INT 21H ;Call DOS

The use of EQU to define the operations makes a program more readable:

CR EQU 13 or EQU 0DH
LF EQU 10 or EQU 0AH
TAB EQU 09 or EQU 09H
MESSAGE DB TAB, 'PC Users Group Annual Report', CR, LF

KEY POINTS TO REMEMBER

� INT 10H is the instruction that links to BIOS for keyboard and display
operations. INT 21H is a special DOS operation that handles some of the
complexity of input/output.

� Be consistent in using hex notation. For example, INT 21 is not the
same as INT 21H.

� Be careful to enter the correct values in the AX, BX, CX, and DX
registers depending on the operation.

� When using INT 21H for original DOS, define a delimiter ($) immediately
following the display area. Be careful when clearing the field not to
clear the delimiter as well. A missing delimiter can cause spectacular
effects on a screen.

� For input under original DOS, define a parameter list carefully. The
INT 21H operation expects the first byte to contain a maximum value and
automatically inserts an actual value in the second byte.

� For extended DOS function call to display, set the AH with hex 40 and
use handle 01 in the BX.

� For extended DOS function call to read, set the AH with hex 3F and use
handle 00 in the BX. Entries are followed in the input area by a return
and a form feed character. The operation does not check for entries
that exceed the maximum.

QUESTIONS

8-1. What is the hex value for the bottom rightmost location on a 40-
column screen?

8-2. Code the instructions to set the cursor to row 12, column 8.
8-3. Code the instructions to clear the screen beginning at row 12, column

0, through row 22, column 79.
8-4. Code data items and instructions to display a message "What is the

date (mm/dd/yy)?." Follow the message with a beep sound. Use (a)
original DOS function calls and (b) extended DOS function calls and
file handles.

8-5. Code data items and instructions to accept data from the keyboard
according to the format in Question 8-4. Use (a) original DOS
function calls and (b) extended DOS function calls and file handles.

8-6. What are the standard file handles for keyboard input, normal screen
display, and printer?

8-7. Key in the program in Figure 8-2 with the following changes then



assemble, link, and test: (a) Instead of row 12, center at row 15.
(b) Instead of clearing the entire screen, clear only rows 0 through
15.

8-8. Revise Figure 8-2 for use with extended DOS function calls for input
and display. Assemble, link, and test.


